

# The Remix Programming Language

An introductory programming language


Robert Sheehan

# Ideas

- ◆ A language to help with program design
- ◆ Readable
- ◆ Self documenting
- ◆ Avoid common problems
- ◆ None of these things are entirely possible

# Also inspired by block programming languages

- ◆ One of the great things about block languages is that the verbal syntax is very readable.



- ◆ The same thing in Python is:

```
random.choice(a_list)
```

or

```
a_list[random.randrange(len(a_list))]
```

- ◆ The same thing in Remix is:

```
choose from a-List
```

or

```
a-List { random (a-List length) }
```


# Developing a program

- ◆ The next sequence of slides shows how a programmer can develop a program starting with a pseudocode description of the program.
- ◆ At each step there is a direct translation into Remix as the programmer refines the code.
- ◆ Parameters get generalised when appropriate.

# Draw 36 circling arrows

=====  
Rotating Arrows  
=====

prepare the window  
prepare the arrows  
animate the arrows




# Draw 36 circling arrows

```
=====  
Rotating Arrows  
=====
```

prepare the window

the-Arrows : 36 arrows

animate the-Arrows




# Draw 36 circling arrows

```
=====  
Rotating Arrows  
=====
```

```
prepare the window  
the-Arrows : 36 arrows  
animate the-Arrows
```


36 arrows :  
make a list for the arrows  
repeat 36 times  
    create a new arrow  
    position the arrow  
    add the arrow to the list  
return the list



# Draw N circling arrows

```
=====  
Rotating Arrows  
=====
```

```
prepare the window  
the-Arrows : 36 arrows  
animate the-Arrows
```



N arrows :  
make a list for the arrows  
repeat N times  
    create a new arrow  
    position the arrow  
    add the arrow to the list  
return the list


# Create an arrow

```
=====  
Rotating Arrows  
=====
```

```
prepare the window  
the-Arrows : 36 arrows  
animate the-Arrows
```

N arrows :  
make a list for the arrows  
repeat N times  
    the-Arrow : an arrow  
        position the arrow  
        add the arrow to the list  
    return the list


an arrow :  
make an arrow shape  
set the arrow colour



# The arrow shape and colour

```
=====  
Rotating Arrows  
=====
```

```
prepare the window  
the-Arrows : 36 arrows  
animate the-Arrows
```



N arrows :  
make a list for the arrows  
repeat N times

    the-Arrow : an arrow  
    position the arrow  
    add the arrow to the list  
return the list

an arrow :


    Arrow : a shape from {  
        { 0, 1 }  
        {-2, 2 }  
        { 0,-2 }  
        { 2, 2 }  
    }

    Arrow's Colour : a random colour  
return Arrow

# The arrow list

```
=====  
Rotating Arrows  
=====
```

```
prepare the window  
the-Arrows : 36 arrows  
animate the-Arrows
```



```
N arrows :  
Arrows : {}  
repeat N times  
  the-Arrow : an arrow  
  position the arrow  
  append the-Arrow to Arrows  
return Arrows  
  
an arrow :  
Arrow : a shape from {  
  { 0, 1 }  
  {-2, 2 }  
  { 0,-2 }  
  { 2, 2 }  
}  
Arrow's Colour : a random colour  
return Arrow
```


# Positioning the arrows

N arrows :

```
Arrows : { }
repeat N times
    the-Arrow : an arrow
    position the-Arrow
    append the-Arrow to Arrows
return Arrows
```


position the-Arrow :

```
X : CENTRE-X + 200 × cosine (Angle)
Y : CENTRE-Y - 200 × sine (Angle)
the-Arrow's Position : { X, Y }
the-Arrow's Heading : -Angle
```



# What about the angle?

```
N arrows :  
  Arrows : {}  
  Angle : 0  
repeat N times  
  the-Arrow : an arrow  
  position the-Arrow at Angle  
  add 10 to Angle  
  append the-Arrow to Arrows  
return Arrows
```



```
position the-Arrow at Angle :  
  X : CENTRE-X + 200 × cosine (Angle)  
  Y : CENTRE-Y - 200 × sine (Angle)  
the-Arrow's Position : { X, Y }  
the-Arrow's Heading : -Angle
```

# We need to animate the arrows

```
=====  
Rotating Arrows  
=====
```

prepare the window

the-Arrows : 36 arrows

**place the-Arrows in the window**

animate the-Arrows

The window is the locus of the animation.  
We can create animation layers in the window.  
Before we can do any animation we need to create the window and place the arrows in the window.

# Preparing the window for animation

=====

Rotating Arrows

=====

the-Window : a graphics window  
the-Animation-Layer : the-Window next layer  
the-Arrows : 36 arrows  
place the-Arrows in the-Animation-Layer  
animate the-Arrows

# Animating the arrows

=====

Rotating Arrows

=====

the-Window : a graphics window  
the-Animation-Layer : the-Window next layer  
the-Arrows : 36 arrows  
place the-Arrows in the-Animation-Layer  
animate the-Arrows

animate the-Arrows :

for each arrow in the-Arrows  
get the angle  
change the angle  
position the arrow at angle

# Animating the arrows

=====

Rotating Arrows

=====

the-Window : a graphics window  
the-Animation-Layer : the-Window next layer  
the-Arrows : 36 arrows  
place the-Arrows in the-Animation-Layer  
animate the-Arrows

animate the-Arrows :

for each Arrow in the-Arrows  
Angle : -(Arrow's Heading)  
add 2 to Angle  
position Arrow at Angle

# Animating the arrows

=====

Rotating Arrows

=====

the-Window : a graphics window  
the-Animation-Layer : the-Window next layer  
the-Arrows : 36 arrows  
place the-Arrows in the-Animation-Layer  
animate the-Arrows **in the-Window**

animate the-Arrows **in the-Window** :  
**animate the-Window 180 times at 30 ticks per second**  
for each Arrow in the-Arrows  
Angle : -(Arrow's Heading)  
add 2 to Angle  
position Arrow at Angle

```
using graphics lib
```

```
-----  
Rotating Arrows  
-----
```

```
the-Window : a graphics window  
the-Animation-Layer : the-Window next layer  
the-Arrows : 36 arrows  
place the-Arrows in the-Animation-Layer  
animate the-Arrows in the-Window
```

```
N arrows :
```

```
Arrows : { }  
Angle : 0  
repeat N times  
    the-Arrow : an arrow  
    position the-Arrow at Angle  
    add 10 to Angle  
    append the-Arrow to Arrows  
return Arrows
```

```
an arrow:
```


```
Arrow : a shape from {  
    { 0, 1 }  
    {-2, 2 }  
    { 0,-2 }  
    { 2, 2 }  
}  
Arrow's Colour : a random colour  
return Arrow
```

```
position the-Arrow at Angle :
```

```
X : CENTRE-X + 200 × cosine (Angle degrees as radians)  
Y : CENTRE-Y - 200 × sine (Angle degrees as radians)  
the-Arrow's Position : { X, Y }  
the-Arrow's Heading : -(Angle degrees as radians)
```

```
animate the-Arrows in the-Window :
```

```
animate the-Window 180 times at 30 ticks per second  
    for each Arrow in the-Arrows  
        Angle : -(Arrow's Heading) radians as degrees  
        add 2 to Angle  
        position Arrow at Angle
```



# A similar turtle graphics program - with turtle objects

with turtle graphics

=====

Rotating Arrows

=====

TSpace : turtle space in (the graphics panel)

the-Arrows : a group of 36 turtles

prepare the-Arrows

place the-Arrows group in TSpace

RADIUS : 200

move the-Arrows to start positions

repeat 180 times

    animate the-Arrows

prepare the-Arrows :

    Angle : 0

    for each Arrow in the-Arrows

        Arrow pen up

        Arrow's Colour : a random colour

        Arrow's Heading : Angle increment by 10

move the-Arrows to start positions :

    the-Arrows's Speed : 200

    move the-Arrows RADIUS steps

    turn the-Arrows -90 degrees

animate the-Arrows :

    the-Arrows' Speed : 5

    move the-Arrows  $(2 \times \text{RADIUS} \times \pi \div 180)$  steps

    turn the-Arrows -2 degrees

# Other examples

- ◆ Factorial
- ◆ The rainfall problem
- ◆ Fizz-buzz
- ◆ Insertion sort

# Other examples

- ◆ Factorial
- ◆ The rainfall problem
  - The beautiful factorial program –
- ◆ Fizz-buzz

- ◆ Insertion sort

```
N ! :  
    if (N = 0)  
        return 1  
    N × (N - 1) !
```

```
for each N from 1 to 10  
    N, "! = ", N ! ↴
```

# Other examples

- ◆ Factorial
- ◆ The rainfall problem
- ◆ Fizz-buzz
- ◆ Insertion sort

```
=====  
The rainfall problem.  
=====
```

```
Rain-Fall : { -1, -3, 0, 5, -2, 1.0, 0, -1, 6, 7, 999, -2, 0, 10 }
```

```
Valid-Data : keep X from Rain-Fall where [ X ≥ 0 ] until [ X = 999 ]
```

```
if (no Valid-Data)  
  "No valid data." ↵  
... otherwise  
  "The average rainfall is ", average of Valid-Data ↵
```

# Other examples

- ◆ Factorial
- ◆ The rainfall problem
- ◆ Fizz-buzz
- ◆ Insertion sort

=====

Fizz-Buzz

=====

for each Number from 1 to 100 do  
when

[ Number is divisible by 15 ] do [ "fizz-buzz" \_ ]  
[ Number is divisible by 3 ] do [ "fizz" \_ ]  
[ Number is divisible by 5 ] do [ "buzz" \_ ]  
Number \_

# Other examples

- ◆ Factorial
- ◆ The rainfall problem

- ◆ Fizz-buzz

- ◆ Insertion sort

```
=====  
Insertion Sort  
=====
```

```
Array : { -1, -3, 0, 5, -2, 1.0, 0, -1, 6, 7, 999 }
```

```
insertion sort Array
```

```
insertion sort List :
```

```
    for each Position from 2 to (length of List) do
```

```
        for each This from Position to 2 do
```

```
            swap if element to the left is greater than This in List
```

N.B. transparent short circuit "::"

```
swap if element to the left is greater than This in List ::
```

```
Left : This - 1
```

```
if (List {Left} > List {This})
```

```
    (List {Left}) swap (List {This})
```

```
... otherwise
```

```
    return
```

# One with tests

-----  
Find the middle NUMBER.  
-----

middle of N1 N2 N3 :

when

[ N1 between N2 and N3 ] do [ N1 ]

[ N2 between N1 and N3 ] do [ N2 ]

N3

N1 between N2 and N3 :

((N2 ≤ N1) and (N1 ≤ N3)) or ((N3 ≤ N1) and (N1 ≤ N2))

-----  
test {  
 "Left"  
 middle of 2 1 3  
 ... expected 2,  
 "Centre"  
 middle of 1 2 3  
 ... expected 2,  
 "Right"  
 middle of 1 3 2  
 ... expected 2  
}

# The Remix Language - keywords

- ◆ true
- ◆ false
- ◆ null
- ◆ return
- ◆ redo
- ◆ create
- ◆ extend
- ◆ getter / setter
- ◆ ME / MY
- ◆ ↓ / ~
- ◆ library
- ◆ using

# The Remix Language - function names

- ◆ A collection of *words* and *parameters*
- ◆ A *word* is a sequence of non-CAPITAL Unicode characters surrounded by whitespace
  - e.g. **! what a\* and item √ list ++**  
(a few characters not allowed)
- ◆ A *parameter* is a word with at least one CAPITAL letter
  - ◆ *parameters* can be inside parentheses () or brackets []
- ◆ Any sequence of *words* and *parameters* is a function name
  - ◆ Must be on one line (when defining)
  - ◆ Ends with : (or sometimes ::)

# Example function names

Value1 and Value2 :

not at end of List :

repeat [Block] until [Condition] :

for each #Item from Start to Finish do [Block] :

filter List by #Item where [Condition] :

draw Colour circle of Size at Centre :

Number is not divisible by any in List :

N ! :

#Number ++ :

When defining a function we represent blocks with [ Name ] - parentheses would work too

# The Remix Language - function calls

- ◆ a function call is like a function name but with all of the formal parameters replaced with expressions evaluating to the actual parameters (and no terminating ":"s)
- ◆ Lisp problem of too many parentheses
- ◆ Don't require parentheses if unambiguous
  - ◆ separated by
    - ◆ operator, brace, bracket, comma, colon
    - ◆ but also literal string, numeric or logic values
    - ◆ and single variable names
- ◆ Can only access local variables and parameters and constants (no global variables)
  - ◆ but there are reference parameters, and composite types (list, string, map, object) can be modified in function calls

# Not like Python

What is the output here?

```
def print_a():
    print(A)
```

```
def change_and_print_a():
    A = 4
    print_a()
    print(A)
```

```
A = 3
print_a()
print(A)
change_and_print_a()
print(A)
```

Answer:

3

3

3

4

3

What is the output here?

```
print a :
A ↴
```

```
change and print a :
A : 4
print a
A ↴
```

```
A : 3
print a
A ↴
change and print a
A ↴
```

Answer:

null

3

null

4

3

"A" is null

# The Remix Language - syntax

- ◆ Generally every line is one statement
  - ◆ But a line continues to following lines if brackets/braces/parentheses are not closed
  - ◆ Also an ellipsis ... at the start of a line indicates it is a continuation
  - ◆ And a ":" can be used to terminate a statement allowing multiple statement lines
- ◆ Indentation is meaningful (even more than in Python)

# The Remix Language - indentation

- ◆ Code starting at the left side margin is top-level code
  - ◆ function definitions always start in the top-level (except in libraries)
    - ◆ all function definitions are gathered before any other statements are run (in main program and then when running libraries)
    - ◆ you can put your functions at the top or bottom of your source file (or anywhere in-between)
  - ◆ all other statements at the top-level are executed in order

# The Remix Language – indentation and deferred blocks

- Tabs are used to indent code
- Most indented areas of code are *deferred blocks*. They contain statements which will be evaluated at some later time.
- Blank lines are ignored (the indentation of following lines is still important).

# This is really important

- We can pass blocks of code as parameters to functions in a syntactically trivial manner
- e.g. in the following function call there are two blocks passed as parameters, one **explicit**, one **implicit**

```
starting with [ Size : 100 ] repeat 4 times
  plot Arrow around centre with Size
    Size : Size + 100
```

# Comments

- ◆ Comments are flexible
  - ◆ A line where the first visible character is "-"
  - ◆ Multiline comments start with a "=" at the margin and continue until a closing "=" at the margin
  - ◆ Any ";" outside a string continuing to the end of the line

==== A comment start ===

Inside the comment.

==== The comment end ===

- lists can be comments
- like this

A : 5 ; this is a comment too

# Statements

- ◆ assignment statement

My-Name : "Robert"

- ◆ return statement - can return nothing

return Something

- ◆ redo statement

redo

- ◆ print statement

"My name is ", My-Name, "." ↴

- ◆ expression

- ◆ using library statement

using graphics lib

block of code which uses the library

# Expressions

- ◆ arithmetic - standard mathematical precedence
- ◆ boolean expressions
- ◆ function calls - always have a value
  - ◆ if no explicit return, use the result of the last statement, some statements have the value "null"
- ◆ accessing a list or map element
- Rain-Fall {2}, Person {"name"}
- ◆ a variable, constant or literal: number, string, logic or list/map/block/object value

# Lists

Rain-Fall : { -1, -3, 0, 5, -2, 1.0, 0, 6, 7, 999, -2, 0, 10 }

- Lists hold any collection of types - even blocks  
 $\{ \text{[Ball } \{X\} < \text{Size}], \text{ [Ball } \{X\} > \text{STD-WIDTH} - \text{Size]} \}$
- They expand as required
- Accessed with {index} - from 1

Rain-Fall {3} or Rain-Fall {N}

# Ranges

1 to 10, 5 to -5

- ◆ direction implicit
- ◆ ranges (and lists) have built-in iterators
  - ◆ only one per range/list currently
- ◆ iterator functions (also work on lists, strings, maps and sequence objects)

Iterator : start Range

next Iterator

end Iterator

# Maps

- ◆ A map/hash/dictionary - with keys and values
- ◆ keys are strings
- ◆ The { } are used for lists and maps.
  - ◆ If the elements are
    - ◆ key : value
  - ◆ then Remix creates a map, otherwise it creates a list.

```
A-Shape : {  
    "position" : { 0, 0 }  
    "heading" : 0 ; straight up  
    "size" : 5  
    "colour" : BLACK  
    "points" : { }  
}
```

```
from Shape with Points :  
new-Shape : copy Shape  
new-Shape {"points"} : Points  
new-Shape
```

```
B-Shape : from A-Shape with { { 1, 2 } }
```

# Objects

- ◆ Objectives
  - ◆ make the creation and use of objects as simple as possible
  - ◆ keep it consistent with the rest of the language
- ◆ The inverse of Uniform Function Call Syntax
  - ◆ instead of allowing function calls to use the method calling syntax, make method calls use the function call syntax
  - ◆ at one level this is as simple as passing the object explicitly as a parameter, but this parameter can be anywhere in the parameter list

# Objects

- The *create* function makes objects with fields and methods.
- A constructor is an ordinary function call that returns an object.
- In a method signature the ME or MY parameter represents the object reference.
- Can call methods on the same object from a method with ME referring to this object.
- Parameters can be the same as field names e.g. Name.

```
a person called Name :  
create
```

```
Name : Name
```

```
getter  
Name
```

```
say hello to ME :  
"Hi ", Name ↴
```

```
welcome ME and bye :  
say hello to ME  
"bye now." ↴
```

```
Person : a person called "Mary"  
Person's Name ↴  
say hello to Person  
welcome Person and bye
```

# Getters and Setters

- ◆ object fields are private
- ◆ automatic getters and setters can be specified using *getter*, *setter* or *getter/setter* blocks
- ◆ a getter is called by:  
`the-Object Field`
- ◆ same as:  
`the-Object's Field`
- ◆ a setter is called by:  
`the-Object Field : new value`
- ◆ same as:  
`the-Object's Field : new value`
- ◆ both getters and setters can be defined manually if access control is desired

# Extending

- ◆ Remix does not have inheritance
- ◆ but you can extend existing objects using a similar syntax to object creation
  - ◆ *extend (existing-Object)* instead of *create*
  - ◆ this creates a new object based on the existing one
- ◆ *extend* can add new fields and methods
- ◆ it can also replace existing method definitions
- ◆ in this way you can specialise objects in a way similar to inheritance

```
a person called Name :
```

```
  create
```

```
    Name : Name
```

```
MY introduction :
```

```
  "My name is ", Name, "." ↵
```

```
say hello to ME :
```

```
  "Hi ", Name, ", pleased to meet you." ↵
```

```
a person called Name age Years:
```

```
  extend (a person called Name)
```

```
    Age : Years
```

```
say hello to ME :
```

```
  "Hi ", Name, ", you are ", Age, " years old." ↵
```

```
Mary : a person called "Mary"
```

```
Mary introduction
```

```
say hello to Mary
```

```
Jack : a person called "Jack" age 30
```

```
Jack introduction
```

```
say hello to Jack
```

```
say hello to (a person called "Stu" age 56)
```

Output

My name is Mary.

Hi Mary, pleased to meet you.

My name is Jack.

Hi Jack, you are 30 years old.

Hi Stu, you are 56 years old.

# Blocks and Block Sequences

- ◆ Blocks hold code for later execution - deferred
  - ◆ the basis for control structures
- ◆ Both explicit [ ... ] and implicit with indentation
- ◆ A block sequence is a block with one or more statements (or lines)
  - ◆ the main program and all function/method bodies are block sequences
- ◆ Looping is done by calling **redo** - this takes control back to the start of the block sequence
  - ◆ this is pretty low-level and not intended for beginner use

# Example control structure

- While a "condition" is true keep doing "block"...

```
while [Condition] repeat [Block] :  
    if (not (do Condition))  
        return Last  
    Last : do Block  
    redo
```

- ... and using it

```
X : 2  
Result : while  
    X is divisible by 2  
... repeat  
    add 3 to X  
Result ↴
```

Output: 5

# Transparent functions

- ◆ Because a block is effectively an anonymous function, return inside a block doesn't always mean a return from its enclosing function.
- ◆ We can make this happen using "transparent" functions
  - ◆ this is only used once in the current standard library and should never be necessary in novice code
  - ◆ we declare a transparent function by ending with "`::`"
    - ◆ see the insertion sort example earlier
  - ◆ if you call a transparent function from the top-level this returns back to the level above the top-level which terminates the program

# Reference functions

- or reference parameters - the parameter name starts with "#"
- Another consequence of creating control structures in Remix itself
- e.g. in a for loop function we need to pass a reference to the index or current value so that it can be accessed in the body of the for loop - which is just a block parameter

```
for each #Item in List [Block] :  
  Pos : start List  
  do  
    if (end Pos)  
      return Last  
    #Item : next Pos  
    Last : do Block  
    redo
```

# Another control structure

With a reference parameter

- Create a new list by applying a block to each element of a list.
- Uses the variable name passed as #Item.

apply Block to each #Item from LIST:

```
Result : { }
for each #Item in LIST
    append (do Block) to Result
```

Result

-... and using it

Result : apply [A × A] to each A from (1 to 10)

Result ↴

```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

# The Remix Language - built-in functions and standard lib functions

- ◆ Built-in functions include
  - ◆ mathematical operations
  - ◆ if/do/type/output functions
  - ◆ list/range/map functions
- ◆ Graphics and turtle graphics
  - ◆ graphics library has both Remix and Java functions
  - ◆ turtle graphics library implemented solely in Remix using the graphics library
- ◆ Control structures (except if/otherwise)
  - ◆ are all written in Remix (can be inspected and changed by the user)

# Current implementation

- ◆ Built in Java using ANTLR grammar
- ◆ Currently an interpreter
- ◆ A simple IDE for the language
  - ◆ helps with code layout
  - ◆ basic function completions
  - ◆ built in graphics and text output

# Some research questions

- ◆ Is it helpful or just a bother?
- ◆ Assuming it is helpful:
  - ◆ Does Remix encourage students to write pseudocode or plan their code?
  - ◆ What is the best level to transition natural language statements into "code" like statements?
  - ◆ What is the best way to convey information about parameter requirements in the function name?
  - ◆ What style of function name is most readable - e.g. should prepositions and articles be included? (Code in Remix written by a speaker of English contains lots of prepositions in function names.)
  - ◆ How applicable is this approach to programmers from different languages, Te Reo?
    - ◆ function call structure is completely flexible
  - ◆ Does Remix make plagiarism checking simpler/more reliable?